Problem:
Score : 400 points

Problem Statement

Akari has n kinds of flowers, one of each kind.

She is going to choose one or more of these flowers to make a bouquet.

However, she hates two numbers a and b, so the number of flowers in the bouquet cannot be a or b.

How many different bouquets are there that Akari can make?

Find the count modulo (10^9 + 7).

Here, two bouquets are considered different when there is a flower that is used in one of the bouquets but not in the other bouquet.

Constraints

All values in input are integers.

2 \leq n \leq 10^9

1 \leq a < b \leq \textrm{min}(n, 2 \times 10^5)

Input

Input is given from Standard Input in the following format:

n a b

Output

Print the number of bouquets that Akari can make, modulo (10^9 + 7). (If there are no such bouquets, print 0.)

Sample Input 1

4 1 3

Sample Output 1

7

In this case, Akari can choose 2 or 4 flowers to make the bouquet.

There are 6 ways to choose 2 out of the 4 flowers, and 1 way to choose 4, so there are a total of 7 different bouquets that Akari can make.

Sample Input 2

1000000000 141421 173205

Sample Output 2

34076506

Print the count modulo (10^9 + 7).

Solution in Haskell (GHC 7.10.3) (judge verdict: Accepted)

import Data.List
import Data.Int

modulus :: Int64
modulus = 10^9 + 7

addMod, subMod, mulMod :: Int64 -> Int64 -> Int64
addMod x y
  | x + y >= modulus = x + y - modulus
  | otherwise        = x + y

subMod x y
  | x - y < 0 = x - y + modulus
  | otherwise = x - y

mulMod x y = (x * y) `rem` modulus

powMod :: Int64 -> Int64 -> Int64
powMod n p
  | p == 0    = 1
  | odd p     = n * powMod n (p-1) `mod` modulus
  | otherwise = let t = powMod n (p `div` 2)
                 in (t^2) `mod` modulus

modInv :: Int64 -> Int64
modInv n = powMod n (modulus - 2)

productMod :: [Int64] -> Int64
productMod = foldr mulMod 1

nCrMod :: Int64 -> Int64 -> Int64
nCrMod n r = productMod [(n-r+1)..n] `mulMod` modInv (productMod [1..r])

main :: IO ()
main = do
    [n,a,b] <- map read . words <$> getLine :: IO [Int64]
    print $ (powMod 2 n - 1) `subMod` nCrMod n a `subMod` nCrMod n b